anchor.diff:    ENDMETHOD.
   
   ENDCLASS.
  +
  +*----------------------------------------------------------------------*
  +*       CLASS lcl_object_iatu DEFINITION
  +*----------------------------------------------------------------------*
  +*
  +*----------------------------------------------------------------------*
  +CLASS lcl_object_iatu DEFINITION INHERITING FROM lcl_objects_super FINAL.
  +
  +  PUBLIC SECTION.
  +    INTERFACES lif_object.
  +
  +  PRIVATE SECTION.
  +    METHODS:
  +      read
  +        EXPORTING es_attr   TYPE w3tempattr
  +                  ev_source TYPE string
  +        RAISING   lcx_exception,
  +      save
  +        IMPORTING is_attr   TYPE w3tempattr
  +                  iv_source TYPE string
  +        RAISING   lcx_exception.
  +
  +ENDCLASS.
  +
  +*----------------------------------------------------------------------*
  +*       CLASS lcl_object_iatu IMPLEMENTATION
  +*----------------------------------------------------------------------*
  +*
  +*----------------------------------------------------------------------*
  +CLASS lcl_object_iatu IMPLEMENTATION.
  +
  +  METHOD read.
  +
  +    DATA: li_template TYPE REF TO if_w3_api_template,
  +          lt_source   TYPE w3htmltabtype,
  +          ls_name     TYPE iacikeyt.
  +
  +
  +    ls_name = ms_item-obj_name.
  +
  +    cl_w3_api_template=>if_w3_api_template~load(
  +      EXPORTING
  +        p_template_name     = ls_name
  +      IMPORTING
  +        p_template          = li_template
  +      EXCEPTIONS
  +        object_not_existing = 1
  +        permission_failure  = 2
  +        error_occured       = 3
  +        OTHERS              = 4 ).
  +    IF sy-subrc <> 0.
  +      _raise 'error from w3api_template~load'.
  +    ENDIF.
  +
  +    li_template->get_attributes( IMPORTING p_attributes = es_attr ).
  +
  +    CLEAR: es_attr-chname,
  +           es_attr-tdate,
  +           es_attr-ttime,
  +           es_attr-devclass.
  +
  +    li_template->get_source( IMPORTING p_source = lt_source ).
  +
  +    CONCATENATE LINES OF lt_source INTO ev_source RESPECTING BLANKS.
  +
  +  ENDMETHOD.
  +
  +  METHOD lif_object~serialize.
  +
  +    DATA: ls_attr   TYPE w3tempattr,
  +          lv_source TYPE string,
  +          lo_xml    TYPE REF TO lcl_xml.
  +
  +
  +    IF lif_object~exists( ) = abap_false.
  +      RETURN.
  +    ENDIF.
  +
  +    read( IMPORTING es_attr   = ls_attr
  +                    ev_source = lv_source ).
  +
  +    CREATE OBJECT lo_xml.
  +    lo_xml->structure_add( ls_attr ).
  +    mo_files->add_xml( lo_xml ).
  +
  +    mo_files->add_html( lv_source ).
  +
  +  ENDMETHOD.
  +
  +  METHOD save.
  +
  +    DATA: lt_source   TYPE w3htmltabtype,
  +          lv_source   TYPE string,
  +          li_template TYPE REF TO if_w3_api_template.
  +
  +
  +    cl_w3_api_template=>if_w3_api_template~create_new(
  +      EXPORTING p_template_data = is_attr
  +                p_program_name = is_attr-programm
  +      IMPORTING p_template = li_template ).
  +
  +    li_template->set_attributes( is_attr ).
  +
  +    lv_source = iv_source.
  +    WHILE strlen( lv_source ) >= 255.
  +      APPEND lv_source(255) TO lt_source.
  +      lv_source = lv_source+255.
  +    ENDWHILE.
  +    IF NOT lv_source IS INITIAL.
  +      APPEND lv_source TO lt_source.
  +    ENDIF.
  +
  +    li_template->set_source( lt_source ).
  +
  +    li_template->if_w3_api_object~save( ).
  +
  +  ENDMETHOD.
  +
  +  METHOD lif_object~deserialize.
  +
  +    DATA: ls_attr   TYPE w3tempattr,
  +          lv_source TYPE string,
  +          lo_xml    TYPE REF TO lcl_xml.
  +
  +
  +    lo_xml = mo_files->read_xml( ).
  +
  +    lo_xml->structure_read( CHANGING cg_structure = ls_attr ).
  +
  +    lv_source = mo_files->read_html( ).
  +
  +    ls_attr-devclass = iv_package.
  +    save( is_attr   = ls_attr
  +          iv_source = lv_source ).
  +
  +  ENDMETHOD.
  +
  +  METHOD lif_object~delete.
  +
  +    DATA: li_template TYPE REF TO if_w3_api_template,
  +          ls_name     TYPE iacikeyt.
  +
  +
  +    ls_name = ms_item-obj_name.
  +
  +    cl_w3_api_template=>if_w3_api_template~load(
  +      EXPORTING
  +        p_template_name     = ls_name
  +      IMPORTING
  +        p_template          = li_template
  +      EXCEPTIONS
  +        object_not_existing = 1
  +        permission_failure  = 2
  +        error_occured       = 3
  +        OTHERS              = 4 ).
  +    IF sy-subrc <> 0.
  +      _raise 'error from if_w3_api_template~load'.
  +    ENDIF.
  +
  +    li_template->if_w3_api_object~set_changeable( abap_true ).
  +    li_template->if_w3_api_object~delete( ).
  +    li_template->if_w3_api_object~save( ).
  +
  +  ENDMETHOD.
  +
  +  METHOD lif_object~exists.
  +
  +    DATA: ls_name TYPE iacikeyt.
  +
  +
  +    ls_name = ms_item-obj_name.
  +
  +    cl_w3_api_template=>if_w3_api_template~load(
  +      EXPORTING
  +        p_template_name     = ls_name
  +      EXCEPTIONS
  +        object_not_existing = 1
  +        permission_failure  = 2
  +        error_occured       = 3
  +        OTHERS              = 4 ).
  +    IF sy-subrc = 1.
  +      rv_bool = abap_false.
  +    ELSEIF sy-subrc <> 0.
  +      _raise 'error from w3_api_template~load'.
  +    ELSE.
  +      rv_bool = abap_true.
  +    ENDIF.
  +
  +  ENDMETHOD.
  +
  +  METHOD lif_object~jump.
  +    _raise 'todo, IATU, jump'.
  +  ENDMETHOD.
  +
  +ENDCLASS.
  +
   
   *----------------------------------------------------------------------*
   *       CLASS lcl_object_dtel DEFINITION
  
positive.code: * See https://github.com/larshp/abapGit/
  
  CONSTANTS: gc_xml_version  TYPE string VALUE 'v0.2-alpha',  "#EC NOTEXT
             gc_abap_version TYPE string VALUE 'v0.103'.      "#EC NOTEXT
  
  ********************************************************************************
  * The MIT License (MIT)
  
negative.code:   METHOD lif_object~serialize.
  
      DATA: lo_xml         TYPE REF TO lcl_xml,
            lv_description TYPE aab_id_descript.
  
  
      IF lif_object~exists( ) = abap_false.
  